
Change 1:
	Function name
	f_IMS_PTC_ImsInfo_Init

	Reason for change
	The implementation of prose CR R5-131292 on 34.229-1 was intended to mandate the use of TCP for 8.x series test cases from 34.229-1. However the regular expression used also affects the operation of 8.* test cases from 36.523-1, which use the same implementation of f_IMS_PTC_ImsInfo_Init().


	Summary of change
	Changed the pattern from  “TC_8_*” to “TC_8_[0-9]#(1,2)” in function f_IMS_PTC_ImsInfo_Init. This should ensure that only 8.* series test case names from 34.226 are matched.

	TTCN module
	IMS_Component.ttcn


Before change

      function f_IMS_PTC_ImsInfo_Init(IP_AddrInfo_Type p_UE_Address,

                                  IP_AddrInfo_Type p_NW_Address,

                                  template (omit) REGISTER_Request p_RegisterReq := omit,

                                  boolean p_SigCompStarted := false) runs on IMS_PTC

  { /* @sic R5s130756 change 14, 15 sic@ */

    var REGISTER_Request v_RegisterReq;

    var InternetProtocol_Type v_TransportProtocol := udp;    /* @sic acc. to discussions regarding rejection of R5-130765 at RAN WG5 Meeting #58 in Malta per default UDP is used for all requests in DL sic@ */

    if (match(testcasename(), pattern "TC_8_*")) {           /* @sic R5-131292 - TCP as normative DL transport protocol in IMS Registration test sic@ */

      v_TransportProtocol := tcp;

    }

    vc_IMS_Global.ImsInfo.UE_Address           := p_UE_Address;

    vc_IMS_Global.ImsInfo.NW_Address           := p_NW_Address;

    vc_IMS_Global.ImsInfo.TransportProtocol    := v_TransportProtocol;

    vc_IMS_Global.ImsInfo.SigCompStarted       := p_SigCompStarted;

    if (ispresent(p_RegisterReq)) {

      v_RegisterReq := valueof(p_RegisterReq);

      vc_IMS_Global.ImsInfo.UriInfo.Registration := f_SIP_SipUrl_GetPublicUserIdentity(f_Addr_Union_GetSipUrl(v_RegisterReq.msgHeader.toField.addressField));

      vc_IMS_Global.ImsInfo.ContactUrl           := f_MessageHeader_GetContactSipUrl(v_RegisterReq.msgHeader);

      vc_IMS_Global.ImsInfo.Cseq.Register        := v_RegisterReq.msgHeader.cSeq.seqNumber;

      vc_IMS_Global.ImsInfo.Registration.CallId  := v_RegisterReq.msgHeader.callId.callid;

    }

  }
After change

  function f_IMS_PTC_ImsInfo_Init(IP_AddrInfo_Type p_UE_Address,

                                  IP_AddrInfo_Type p_NW_Address,

                                  template (omit) REGISTER_Request p_RegisterReq := omit,

                                  boolean p_SigCompStarted := false) runs on IMS_PTC

  { /* @sic R5s130756 change 14, 15 sic@ */

    var REGISTER_Request v_RegisterReq;

    var InternetProtocol_Type v_TransportProtocol := udp;    /* @sic acc. to discussions regarding rejection of R5-130765 at RAN WG5 Meeting #58 in Malta per default UDP is used for all requests in DL sic@ */

      if (match(testcasename(), pattern "TC_8_[0-9]#(1,2)")) {           /* @sic R5-131292 - TCP as normative DL transport protocol in IMS Registration test sic@ */

      v_TransportProtocol := tcp;

    }

    vc_IMS_Global.ImsInfo.UE_Address           := p_UE_Address;

    vc_IMS_Global.ImsInfo.NW_Address           := p_NW_Address;

    vc_IMS_Global.ImsInfo.TransportProtocol    := v_TransportProtocol;

    vc_IMS_Global.ImsInfo.SigCompStarted       := p_SigCompStarted;

    if (ispresent(p_RegisterReq)) {

      v_RegisterReq := valueof(p_RegisterReq);

      vc_IMS_Global.ImsInfo.UriInfo.Registration := f_SIP_SipUrl_GetPublicUserIdentity(f_Addr_Union_GetSipUrl(v_RegisterReq.msgHeader.toField.addressField));

      vc_IMS_Global.ImsInfo.ContactUrl           := f_MessageHeader_GetContactSipUrl(v_RegisterReq.msgHeader);

      vc_IMS_Global.ImsInfo.Cseq.Register        := v_RegisterReq.msgHeader.cSeq.seqNumber;

      vc_IMS_Global.ImsInfo.Registration.CallId  := v_RegisterReq.msgHeader.callId.callid;

    }

}

Change 2:

	Function name
	f_IMS_EmergencyCallSetup_AnnexC22

	Reason for change
	From test spec, 34.229-1 Anx. A.2.7 ACK; route-param will have the same values as Record-Route header of 183, 180 or 200 response in reverse order. As per test case "Emergency call with emergency registration" function "f_IMS_AckRequest_MessageHeaderRX" needs to be modified to pass the 3rd parameter to function "f_IMS_RouteSet_MO_Call_TX" to handle the emergency call procedure.

	Summary of change
	3rd parameter for function "f_IMS_AckRequest_MessageHeaderRX" is passed as True.

	TTCN module
	IMS_Procedures_CallControl.ttcn


Before change

    function f_IMS_EmergencyCallSetup_AnnexC22(INVITE_A_2_1_Context_Type p_Context := A_2_1_A7,

                                             boolean p_GeolocationInfoAvailable := false) runs on IMS_PTC return INVITE_Request

  {
 …………
    // Step 5. Finally, UE acknowledges call setup.

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);

    IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeContactSipUri, v_MessageHeader_Ack)));

    return v_InviteRequest;

  }
After change

    function f_IMS_EmergencyCallSetup_AnnexC22(INVITE_A_2_1_Context_Type p_Context := A_2_1_A7,

                                             boolean p_GeolocationInfoAvailable := false) runs on IMS_PTC return INVITE_Request

  { 
…………

    // Step 5. Finally, UE acknowledges call setup.

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200,true);

    IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeContactSipUri, v_MessageHeader_Ack)));

    return v_InviteRequest;

  }
Change 3:

	Function name
	f_IMS_AckRequest_MessageHeaderRX

	Reason for change
	1. From test spec, 34.229-1 Anx. A.2.7 ACK; route-param will have the   same values as Record-Route header of 183, 180 or 200 response in reverse order. As per test case "Emergency call with emergency registration" function "f_IMS_AckRequest_MessageHeaderRX" needs to be modified to pass the 3rd parameter.

2. The 3rd parameter will be passed to function "f_IMS_RouteSet_MO_Call_TX" to handle the emergency call procedure.

	Summary of change
	1. boolean parameter "p_IsEmergency" is added in the function "f_IMS_AckRequest_MessageHeaderRX"
2. parameter "p_IsEmergency" is passed to function "f_IMS_RouteSet_MO_Call_TX"

	TTCN module
	IMS_Procedures_CallControl.ttcn


Before change

      function f_IMS_AckRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                            template (value) StatusLine p_StatusLine_PrevResponse)

    runs on IMS_PTC return template (present) MessageHeader

  { /* @sic R5s130684: p_StatusLine_PrevResponse made mandatory, p_MessageHeader_PrevResponse removed, new parameter p_IsGiba sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Ack := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_MessageHeader_GetCseqValue(v_MessageHeader_Invite);

    var template (value) RouteBody_List v_RouteBodyList

    var integer v_StatusCode := valueof(p_StatusLine_PrevResponse.statusCode);

        if (not (v_StatusCode == 180 or v_StatusCode == 183 or v_StatusCode == 200))

    {

        v_RouteBodyList := v_MessageHeader_Invite.route.routeBody;

    }

    else

    {

        v_RouteBodyList := fl_RouteSet_ReverseOrder(f_IMS_RouteSet_MO_Call_TX());  /* @sic R5s130684: same as for 180 or 183 response in reverse order sic@ */

    }

    if ((v_StatusCode < 200) or (v_StatusCode > 299)) {                                       /* @sic R5-130682: ACK for non-2xx response sic@ */

      v_MessageHeader_Ack.via           := v_MessageHeader_Invite.via;

    } else {

      v_MessageHeader_Ack.via           := fl_Via_AsInPrevInvite(v_MessageHeader_Invite.via); /* @sic R5-130682: ACK for 2xx response sic@ */

    }

    v_MessageHeader_Ack.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Ack.toField         := f_IMS_RequestInDialog_ToHeaderRX(v_MessageHeader_Invite);    /* @sic R5s130497 change 1 - MCC160 implementation; R5-133303 sic@ */

    v_MessageHeader_Ack.fromField       := f_IMS_RequestInDialog_FromHeaderRX(v_MessageHeader_Invite);  /* @sic R5s130497 change 1 - MCC160 implementation; R5-133303 sic@ */

    v_MessageHeader_Ack.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Ack.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Ack.cSeq            := cr_CseqDef(v_CSeqValue, "ACK");

    v_MessageHeader_Ack.maxForwards     := cr_MaxForwardsDef;

    v_MessageHeader_Ack.pAccessNetworkInfo := omit;

    return v_MessageHeader_Ack;

  }

After change

      function f_IMS_AckRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                            template (value) StatusLine p_StatusLine_PrevResponse,
                                            boolean p_IsEmergency := false) 
    runs on IMS_PTC return template (present) MessageHeader

  { /* @sic R5s130684: p_StatusLine_PrevResponse made mandatory, p_MessageHeader_PrevResponse removed, new parameter p_IsGiba sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Ack := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_MessageHeader_GetCseqValue(v_MessageHeader_Invite);

    var template (value) RouteBody_List v_RouteBodyList;

    var integer v_StatusCode := valueof(p_StatusLine_PrevResponse.statusCode);

    if (not (v_StatusCode == 180 or v_StatusCode == 183 or v_StatusCode == 200))

    {

        v_RouteBodyList := v_MessageHeader_Invite.route.routeBody;

    }

    else

    {

        v_RouteBodyList := fl_RouteSet_ReverseOrder(f_IMS_RouteSet_MO_Call_TX(p_IsEmergency));  /* @sic R5s130684: same as for 180 or 183 response in reverse order sic@ */

    }

    if ((v_StatusCode < 200) or (v_StatusCode > 299)) {                                       /* @sic R5-130682: ACK for non-2xx response sic@ */

      v_MessageHeader_Ack.via           := v_MessageHeader_Invite.via;

    } else {

      v_MessageHeader_Ack.via           := fl_Via_AsInPrevInvite(v_MessageHeader_Invite.via); /* @sic R5-130682: ACK for 2xx response sic@ */

    }

    v_MessageHeader_Ack.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Ack.toField         := f_IMS_RequestInDialog_ToHeaderRX(v_MessageHeader_Invite);    /* @sic R5s130497 change 1 - MCC160 implementation; R5-133303 sic@ */

    v_MessageHeader_Ack.fromField       := f_IMS_RequestInDialog_FromHeaderRX(v_MessageHeader_Invite);  /* @sic R5s130497 change 1 - MCC160 implementation; R5-133303 sic@ */

    v_MessageHeader_Ack.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Ack.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Ack.cSeq            := cr_CseqDef(v_CSeqValue, "ACK");

    v_MessageHeader_Ack.maxForwards     := cr_MaxForwardsDef;

    v_MessageHeader_Ack.pAccessNetworkInfo := omit;

    return v_MessageHeader_Ack;

  }
Change 4:

	Function name
	f_IMS_OtherResponse_200_MessageHeader_RX

	Reason for change
	The current TTCN implementation mandates that the UE includes the “record-route” header in the 200 OK response for an UPDATE request. This is not required by the core specifications, according to 24.229 A.2.2.4.14. An accompanying prose CR will be raised at the next RAN5 meeting to adjust 34.229-1 A.3.1.

	Summary of change
	Allow “record-route” header to be omitted in 200 OK response for an UPDATE request.

	TTCN module
	IMS_Procedures_Common.ttcn


Before change

  function f_IMS_OtherResponse_200_MessageHeaderRX(template (value) MessageHeader p_MessageHeader_Request,

                                                   template ContentType p_ContentType := omit) runs on IMS_PTC return template (present) MessageHeader

  { /* NOTE: Message is sent by the UE => conditions A2, A4 and A5 are applicable;

     *       For A4 (GIBA) pAccessNetworkInfo is optional */

    /* @sic R5-130683 sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    var template (present) MessageHeader v_MessageHeader_Response := cr_MessageHeader_Dummy;

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);  /* acc. to A.3.1 condition A5: access network technology and, if applicable, the cell ID */

    var boolean v_A5 := true;    /* true => response sent by the UE within a dialog */

    var boolean v_A2A4 := false; /* true => Response sent by UE for INVITE/UPDATE */

    var charstring v_Method := valueof(p_MessageHeader_Request.cSeq.method);

    var integer v_Port_us;

    select (v_Method) {

      case ("NOTIFY", "MESSAGE") { v_A5 := false; }

      case ("INVITE", “UPDATE”)  { v_A2A4 := true; }

    }

    v_MessageHeader_Response.via          := p_MessageHeader_Request.via;

    v_MessageHeader_Response.fromField    := f_IMS_Response_FromHeaderRX(p_MessageHeader_Request); /* @sic R5s130453 sic@ */

    v_MessageHeader_Response.toField      := f_IMS_Response_ToHeaderRX(p_MessageHeader_Request);   /* @sic R5s130183 change 4 sic@   @sic R5-131897, R5-132063 - new generic function sic@ */

    v_MessageHeader_Response.callId       := p_MessageHeader_Request.callId;

    v_MessageHeader_Response.sessionId    := p_MessageHeader_Request.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Response.cSeq         := p_MessageHeader_Request.cSeq;

    v_MessageHeader_Response.contentType  := p_ContentType;

    if (not f_IMS_PTC_SecurityScheme_IsGiba() and v_A5) {     // @sic R5-130683: pAccessNetworkInfo is optional for GIBA sic@

      v_MessageHeader_Response.pAccessNetworkInfo := v_PAccessNetworkInfo;

    } else {

      v_MessageHeader_Response.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

    }

    if (v_A2A4) {

      v_Port_us := f_IMS_PTC_GetPort_us();

      v_MessageHeader_Response.recordRoute := cr_RecordRoute(fl_IMS_RouteSet_MT_Call_RX());      // !!!! to be replaced by cr_RecordRoute(f_IMS_RouteSet_MT_Call_TX()) !!!!

      v_MessageHeader_Response.contact     := cr_Contact(cr_SipUri_HostPort(?, v_Port_us));    // host address needs to be checked after receiving

    }

    return v_MessageHeader_Response;

  }

}

After change

     function f_IMS_OtherResponse_200_MessageHeaderRX(template (value) MessageHeader p_MessageHeader_Request,

                                                   template ContentType p_ContentType := omit) runs on IMS_PTC return template (present) MessageHeader

  { /* NOTE: Message is sent by the UE => conditions A2, A4 and A5 are applicable;

     *       For A4 (GIBA) pAccessNetworkInfo is optional */

    /* @sic R5-130683 sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    var template (present) MessageHeader v_MessageHeader_Response := cr_MessageHeader_Dummy;

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);  /* acc. to A.3.1 condition A5: access network technology and, if applicable, the cell ID */

    var boolean v_A5 := true;    /* true => response sent by the UE within a dialog */

    var boolean v_A2A4 := false; /* true => Response sent by UE for INVITE/UPDATE */

    var charstring v_Method := valueof(p_MessageHeader_Request.cSeq.method);

    var integer v_Port_us;

    select (v_Method) {

      case ("NOTIFY", "MESSAGE") { v_A5 := false; }

      case ("INVITE”)  { v_A2A4 := true; }

    }

    v_MessageHeader_Response.via          := p_MessageHeader_Request.via;

    v_MessageHeader_Response.fromField    := f_IMS_Response_FromHeaderRX(p_MessageHeader_Request); /* @sic R5s130453 sic@ */

    v_MessageHeader_Response.toField      := f_IMS_Response_ToHeaderRX(p_MessageHeader_Request);   /* @sic R5s130183 change 4 sic@   @sic R5-131897, R5-132063 - new generic function sic@ */

    v_MessageHeader_Response.callId       := p_MessageHeader_Request.callId;

    v_MessageHeader_Response.sessionId    := p_MessageHeader_Request.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Response.cSeq         := p_MessageHeader_Request.cSeq;

    v_MessageHeader_Response.contentType  := p_ContentType;

    if (not f_IMS_PTC_SecurityScheme_IsGiba() and v_A5) {     // @sic R5-130683: pAccessNetworkInfo is optional for GIBA sic@

      v_MessageHeader_Response.pAccessNetworkInfo := v_PAccessNetworkInfo;

    } else {

      v_MessageHeader_Response.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

    }

    if (v_A2A4) {

      v_Port_us := f_IMS_PTC_GetPort_us();

      v_MessageHeader_Response.recordRoute := cr_RecordRoute(fl_IMS_RouteSet_MT_Call_RX());      // !!!! to be replaced by cr_RecordRoute(f_IMS_RouteSet_MT_Call_TX()) !!!!

      v_MessageHeader_Response.contact     := cr_Contact(cr_SipUri_HostPort(?, v_Port_us));    // host address needs to be checked after receiving

    }

    return v_MessageHeader_Response;

  }

}
Change 5

	Function name
	Fl_EUTRA_IPCAN_ActivateDedicatedEpsBearer_SpeechCall()

	Reason for change
	1. According to 34.229-3 8.1.3, the EPS bearer set up for an IMS speech call should use dedicated bearer context#1 as defined in 36.508 6.6.2. This should actually be dedicated bearer context#4, since bearer context#1 omits the traffic filter template to ensure RTP traffic goes over this bearer.

2. The generic procedures in 36.508 for establishing IMS speech calls (36.508 4.5A.6.3 and 4.5A.7.3) state that an RLC-UM data radio bearer is associated with the dedicated EPS bearer used to carry IMS speech. The current implementation for the multi-tester model configures an RLC AM bearer by default.

	Summary of change
	1. In function fl_EUTRA_IPCAN_ActivateDedicatedEpsBearer_SpeechCall(), use bearer context id#4 when configuring QoS and TFT for the dedicated EPS bearer. An associated prose CR to correct 34.229-3 will be raised.
2. Specify RLC UM mode when creating the associated EUTRA DRB for the dedicated EPS bearer.

	TTCN module
	IPCAN_EUTRA.ttcn


Before change

  function fl_EUTRA_IPCAN_ActivateDedicatedEpsBearer_SpeechCall(EUTRA_CellId_Type p_CellId,

                                                                DRB_Identity p_DrbId := tsc_DRB2,

                                                                EPS_BearerIdentity p_DedicatedEpsBeareId := tsc_EpsDedicatedBearerId) runs on EUTRA_PTC

  {

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(p_CellId,

                                        p_DrbId,

                                        p_DedicatedEpsBeareId,

                                        v_EPS_TI,

                                        cs_LinkedId(hex2bit(tsc_EpsDefaultBearerId)),

                                        cs_508_EPS_QoS_Dedicated_1,

                                        f_EUTRA_Get_508_TrafficFlowTemplate(1, p_DedicatedEpsBeareId));
    //The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message

    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(p_DedicatedEpsBeareId))));

  }
After change

     function fl_EUTRA_IPCAN_ActivateDedicatedEpsBearer_SpeechCall(EUTRA_CellId_Type p_CellId,

                                                                DRB_Identity p_DrbId := tsc_DRB2,

                                                                EPS_BearerIdentity p_DedicatedEpsBeareId := tsc_EpsDedicatedBearerId) runs on EUTRA_PTC

  {

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(p_CellId,

                                        p_DrbId,

                                        p_DedicatedEpsBeareId,

                                        v_EPS_TI,

                                        cs_LinkedId(hex2bit(tsc_EpsDefaultBearerId)),

                                        cds_EPS_QoS_Dedicated_4,

                                        f_EUTRA_Get_508_TrafficFlowTemplate(4, p_DedicatedEpsBeareId),UM_Mode);
    //The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message

    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(p_DedicatedEpsBeareId))));

  }
Change 6:

	Function name
	f_IMS_BuildSDP_Annex11_Step4or8

	Reason for change
	RFC 3264 section 5.1 indicates that the RTP payload type numbers received in an SDP answer from a UE may not match the payload types sent in the initial SDP offer:
However, for sendonly and sendrecv streams, the answer might indicate

different payload type numbers for the same codecs, in which case,

the offerer MUST send with the payload type numbers from the answer.

For MT call scenarios, TTCN currently sends ‘97’ at the payload type for AMR 8000 in the INIVITE SDP, and expects to receive ‘97’ in the answer.



	Summary of change
	Use default parameters for RTP payload type when building the expected SDP answer at steps 4 and 8 of Annex C21. The templates will then use the pattern tsc_SDP_AnyMediaFormatDescr.

	TTCN module
	IMS_SDP_Messages.ttcn


Before change

  function f_IMS_BuildSDP_AnnexC11_Step4or8(charstring p_Step) runs on IMS_PTC return template (present) SDP_Message

  { /* @sic R5-123077 change 3: 'inactive' removed from 'Attributes for media' for step 4 sic@ */

    //var charstring v_FmtAudio := tsc_SDP_AnyMediaFormatDescr; //shaun tsc_SDP_FmtAudio; 

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrConnection := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc;

    var template (present) SDP_attribute_list v_SDP_Attribute_List;

    select (p_Step) {

      case ("step 4") {

        v_SDP_Attribute_List := {

          cr_SDP_Attribute_rtpmap(v_FmtAudio, tsc_CODEC_AMR_8000_RX_Pattern),     // @sic R5s130681 change 2 sic@

          cr_SDP_Attribute_fmtp(v_FmtAudio),

          // precondition attributes:

          cr_SDP_Attribute_curr_quos(c_local, (c_none, c_sendrecv)),     // @sic R5s120530 change 3.15 sic@

          cr_SDP_Attribute_curr_quos(c_remote, c_none),

          cr_SDP_Attribute_des_quos(c_mandatory, c_local),

          cr_SDP_Attribute_des_quos(c_mandatory, c_remote),

          *

        };

      }

      case ("step 8") {

        v_SDP_Attribute_List := {

          // media attributes:

          cr_SDP_Attribute_rtpmap(v_FmtAudio, tsc_CODEC_AMR_8000_RX_Pattern),     // @sic R5s130681 change 2 sic@

          cr_SDP_Attribute_fmtp(v_FmtAudio),

          cr_SDP_Attribute_sendrecv,

          // precondition attributes:

          cr_SDP_Attribute_curr_quos(c_local, c_sendrecv),

          cr_SDP_Attribute_curr_quos(c_remote, c_sendrecv),

          cr_SDP_Attribute_des_quos(c_mandatory, c_local),

          cr_SDP_Attribute_des_quos(c_mandatory, c_remote),

          *

        };

      }

      case else {

        FatalError(__FILE__, __LINE__, "");

      }

    }

    v_SDP_Media_Desc := cr_SDP_Media_Desc(cr_SDP_Media_Audio(v_FmtAudio),

                                          cr_SDP_Connection(v_AddrType, v_IPAddrConnection),

                                          cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                          v_SDP_Attribute_List);

    return cr_SDP_Message_Def(v_AddrType, v_IPAddrOrigin, v_IPAddrConnection, v_SDP_Media_Desc);

  }
After change

function f_IMS_BuildSDP_AnnexC11_Step4or8(charstring p_Step) runs on IMS_PTC return template (present) SDP_Message

  { /* @sic R5-123077 change 3: 'inactive' removed from 'Attributes for media' for step 4 sic@ */

    //var charstring v_FmtAudio := tsc_SDP_AnyMediaFormatDescr; //shaun tsc_SDP_FmtAudio; 

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrConnection := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc;

    var template (present) SDP_attribute_list v_SDP_Attribute_List;

    select (p_Step) {

      case ("step 4") {

        v_SDP_Attribute_List := {

          cr_SDP_Attribute_rtpmap(-, tsc_CODEC_AMR_8000_RX_Pattern),     // @sic R5s130681 change 2 sic@

          cr_SDP_Attribute_fmtp(),

          // precondition attributes:

          cr_SDP_Attribute_curr_quos(c_local, (c_none, c_sendrecv)),     // @sic R5s120530 change 3.15 sic@

          cr_SDP_Attribute_curr_quos(c_remote, c_none),

          cr_SDP_Attribute_des_quos(c_mandatory, c_local),

          cr_SDP_Attribute_des_quos(c_mandatory, c_remote),

          *

        };

      }

      case ("step 8") {

        v_SDP_Attribute_List := {

          // media attributes:

          cr_SDP_Attribute_rtpmap(-, tsc_CODEC_AMR_8000_RX_Pattern),     // @sic R5s130681 change 2 sic@

          cr_SDP_Attribute_fmtp(),

          cr_SDP_Attribute_sendrecv,

          // precondition attributes:

          cr_SDP_Attribute_curr_quos(c_local, c_sendrecv),

          cr_SDP_Attribute_curr_quos(c_remote, c_sendrecv),

          cr_SDP_Attribute_des_quos(c_mandatory, c_local),

          cr_SDP_Attribute_des_quos(c_mandatory, c_remote),

          *

        };

      }

      case else {

        FatalError(__FILE__, __LINE__, "");

      }

    }

    v_SDP_Media_Desc := cr_SDP_Media_Desc(cr_SDP_Media_Audio(),

                                          cr_SDP_Connection(v_AddrType, v_IPAddrConnection),

                                          cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                          v_SDP_Attribute_List);

    return cr_SDP_Message_Def(v_AddrType, v_IPAddrOrigin, v_IPAddrConnection, v_SDP_Media_Desc);

  }
